Chapitre 1

Introduction a Prolog

1.1 Introduction

Prolog est un langage de programmation basé sur la logique du premier ordre, il a été
inventé au début des années 70 par Alain Colmerauer a Marseille justement dans le but de
pouvoir faire du traitement de la langue naturelle, mais il s’est vite apercu que ce langage
pouvait avoir un champ d’application beaucoup plus large.

Cette section introduit tres brievement Prolog en utilisant des concepts et des exemples
extraits du domaine du traitement de la langue naturelle. Nous employons la syntaxe dite
d’Edimbourg (Clocksin Mellish 84), (Saint-Dizier 87), (Sterling et Shapiro 86),

Un programme Prolog se présente comme une suite de regles ou clauses de la forme
suivante :

P :-Q1, Q2, ... , Qn.

qu’on interprete comme :
P est vrai si Q1,Qo, ..., Q, sont vrais.

Une regle de la forme
P.

est appelé un fait car il n’y a aucune condition pour sa véracité. P est appelé la téte et
Q1,Qs,...,Q, le corps de la regle. Chacun de ces éléments est appelé un littéral composé
d’un symbole de prédicat et de paramétres ou arguments entre parentheses séparés par des
virgules. Par exemple,

syntagmeNominal adverbe (X) pronom(je,1,Nombre)

sont des littéraux indiquant des relations entre les arguments. Les parametres sont des termes
composés de variables dénotant des objets a définir plus tard, des symboles atomiques ou
des termes composés. Revenons sur ces notions :

variables notées par des identificateurs débutant par une majuscule ou un souligné; par
exemple :

X Genre _nombre

Une variable notée par un seul souligné indique que la valeur de la variable ne nous
importe pas.

symboles atomiques notés soit par des nombres ou des identificateurs débutant par une
minuscule.

325 3.1416 pluriel masculin

termes composés notés par un foncteur et des parametres qui sont eux-mémes des termes,
par exemple :

sn(X,Y)  sv(vb(mange),sn(det(la),nom(souris))) +(*(3,7),8)

Ils représentent des arbres ou le foncteur étiquette la racine et les parametres les
branches. Les arbres qui correspondent aux exemples précédents sont donc :

X Y \Y S
AN A
mange dft nTm 3 7

la  souris

1.2 Notion de programme Prolog

Un programme Prolog est composé d’une suite de faits et de relations entre ces faits
exprimées par des regles selon la syntaxe introduite dans la section précédente. Soit la liste
suivante indiquant quelques faits du vocabulaire francais :

determinant(la).
determinant(le).

nom(souris) .
nom(chat) .

adjectif (blanc).
adjectif (rouge) .
adjectif (blanche).

genre(la,feminin) .
genre(le,masculin) .
genre(souris,feminin).
genre(chat,masculin).
genre(blanc,masculin) .
genre(blanche,feminin) .
genre(rouge,_) .

L’utilisation d’une variable dans le dernier fait signifie que sa valeur ne nous importe pas.
Les regles indiquent des relations entre des faits; par exemple :

accord(X,Y) :- genre(X,Z), genre(Y,Z).

dit que X est du méme genre que Y si le genre de X est d’un certain Z et ce Z est également le
genre de Y. [’affectation a une variable Prolog est unique et ne se fait que par le passage de
parametre. Une fois que sa valeur a été fixée par un fait ou une regle, elle est répercutée a
toutes les autres apparitions dans la regle. Dans I'exemple précédent, aussitot que la premiere
occurrence de Z est fixée par la recherche du genre de X alors la valeur de Z pour Y est fixée
et on vérifiera alors que Y est bien du méme genre que X. On exprime une alternative en
donnant plusieurs possibilités de regles :

sn(X,Y) :-determinant (X), nom(Y), accord(X,Y).
sn(X,Y) :-nom(X), adjectif(Y), accord(X,Y).

indique une relation entre deux individus quelconques (X et Y) disant que le couple X et Y
forment un syntagme correct soit si X est un déterminant et Y un nom, soit X est un nom
et Y est un adjectif. Dans les deux cas, on vérifie I’accord des genres avec le prédicat défini
plus haut. Les faits et les regles illustrés plus haut sont vraiment tres simplistes et ne servent
qu’a illustrer nos premiers exemples ; nous verrons plus loin d’autres méthodes plus efficaces
de travailler.

L’exécution d’un tel programme est lancée par une question (ou but) posée a l'interprete
Prolog qui va vérifier si ce qu’on demande est vrai ; cette question est donnée apres l'incitation
qui est :
| 7-

En Sicstus Prolog qui nous a servi comme outil de test des programmes donnés ici. Par
exemple!,

| ?- nom(chat).

yes
| ?- nom(chien).

no
| 7- accord(le,chat).

yes
| ?- accord(chat,souris).

no
| 7-
Le systeme répond par yes s’il trouve une fagon de prouver le nouveau fait a partir des faits
et des relations donnés dans le programme. Par exemple, le premier fait a prouver est vrai
car il est présent dans notre liste de faits alors que le second n’y est pas. Dans le troisieme
cas, on utilise la regle accord pour trouver que les deux arguments sont du méme genre alors
que le quatrieme échoue car les mots chat et souris ne sont pas du méme genre (et non pas
pour des raisons sémantiques). Dans le cas ol on a mis des variables dans la question, on
recoit les valeurs des variables permettant de prouver le fait qu’on demande. Par exemple :

| 7- genre(souris,X).
X = feminin

'Dans les exemples d’interactions, ce qui est précédé de ?- correspond aux messages de I'usager et les
lignes ne débutant pas par ce symbole sont les réponses du systeme



4 CHAPITRE 1. INTRODUCTION A PROLOG

yes

I'interprete demande alors si on désire les autres possibilités, si oui, on répond par ; et il
tente de trouver une autre affectation des variables rendant ce fait vrai. Si on répond autre
chose, alors le systeme répond yes pour indiquer qu’il y avait peut-étre d’autres solutions.
Pour obtenir tous les mots masculins dans notre banque, il suffit de faire :

| 7- genre(Mot,masculin).

Mot = le ;

Mot = chat ;

Mot = blanc ;

Mot = rouge ;
no

le systeme répond par no lorsque toutes les possibilités ont été épuisées. Il faut remarquer
que rouge est donné en réponse car il peut correspondre a n’importe quelle valeur.

| 7- sn(X,Y).

X = 1la

Y = souris ;

X = le

Y = chat ;

X = souris

Y = rouge ;

X = souris

Y = blanche ;

X = chat

Y = blanc ;

X = chat

Y = rouge ;
no

L’ordre de sortie des résultats dépend de I'ordre d’apparition des faits et des regles. Il corres-
pond a un parcours en profondeur des différentes possibilités : ainsi dans le dernier exemple,
on débute par la premiere possibilité pour un sn soit un déterminant et un nom; pour un
déterminant on commence par le premier la, ensuite on tente de trouver un nom mais du
bon accord. Ensuite, on reprend avec l'autre article jusqu’a épuisement de ces possibilités;
on passe a la deuxieme possibilité de sn : on cherche tous les noms et pour chacun d’eux les
adjectifs du bon genre.

On peut également donner plusieurs buts a résoudre et alors il faut que tous soient
satisfaits ; par exemple, pour obtenir tous les noms masculins de notre dictionnaire :

| 7- genre(Mot,masculin) ,nom(Mot).

Mot = chat ;
no

Le processus qui vérifie la conformité entre les parametres est beaucoup plus général que
celui qu’on rencontre dans la plupart des langages de programmation car les parametres
et les valeurs peuvent contenir des variables qui sont liées aux valeurs correspondant dans
I’autre. Cette opération est appelée unification et est tres pratique pour la construction de
structures durant des processus d’analyse syntaxique. Par exemple :

p(snm(det (X) ,nom(Y),G),X,Y) :-sn(X,Y) ,genre(X,G).
qu’on lance comme suit :

| ?- p(8,le,chat).

S = snm(det(le) ,nom(chat) ,masculin) ;
no

| 7- p(S,X,rouge) .

S = snm(det(souris) ,nom(rouge),feminin)
X = souris ;
S = snm(det(chat) ,nom(rouge) ,masculin)
X = chat ;

no

Dans le premier cas, le genre s’est propagé dans la structure finale méme s’il n’est déterminé
qu’a la fin. Dans le deuxieme cas, on cherche a trouver une structure de sn avec rouge en
deuxieme position. On voit qu’il suffit de définir la forme du foncteur qu’on veut créer pour
que le processus d’unification fasse le nécessaire pour créer la structure.

1.3 Création de programmes

Pour créer une base de faits et de regles, on écrit normalement un fichier dont on demande
le chargement par l'interprete ou le compilateur. Cette opération est appelée consultation
d’un fichier et se commande en faisant exécuter le but

consult(fichier)

qui va interpréter le contenu de fichier comme une suite de clauses a ajouter a la banque de
faits et regles du systeme. Il est souvent pratique de pouvoir remplacer un jeu de clauses par
d’autres du méme nom : c’est ce qui arrive lorsqu’on découvre une erreur dans un fichier,
qu'on la corrige avec I’éditeur et fait interpréter le programme en s’attendant a ce que la
nouvelle version des clauses remplace les anciennes au lieu de s’ajouter aux anciennes qui
sont erronées. Cette opération de remplacement est commandée par

reconsult(fichier)

On peut ajouter des commentaires dans un fichier en les faisant précéder par le caractere % ;
le commentaire s’étend jusqu’a la fin de la ligne.

1.4 Arithmétique

Les foncteurs permettent d’exprimer toutes les valeurs, mais dans ce cas I'arithmétique
devient encombrante et inefficace car pour représenter par exemple 3, il faudrait écrire
succ(succ(succ(0))). Heureusement, on peut effectuer des opérations en utilisant le fonc-
teur is qui évalue son deuxieme argument et 1'unifie avec le premier. Ainsi

is(X, +(x(3,7),8))

unifiera X avec 29, car le deuxieme argument est considéré comme un arbre dont les noeuds
sont soit des valeurs soit des opérations a effectuer sur les branches. C’est déja beaucoup plus
pratique mais 'arithmétique sous forme préfixe n’est pas tres habituelle (a moins d’étre un
lispien convaincu...). C’est pourquoi la plupart de ces foncteurs peuvent aussi s’écrire sous
forme infixe (pour les foncteurs binaires) et sous forme préfixe ou suffixe pour les foncteurs
unaires. L’utilisateur peut également déclarer certain de ses propres foncteurs comme étant
infixe, préfixe ou suffixe. A chaque opérateur sont associées une priorité et une associativité,
ce qui permet de rendre les programmes plus lisibles et reproduire ainsi les priorités ren-
contrées habituellement dans les langages de programmation. L’exemple précédent devient
donc :

X is 3%7+8

qui n’est pas sans rappeler 'affectation classique, mais c’est une illusion car ceci n’est qu’une

autre facon de dénoter I'arbre suivant :
A
A
* 8
3 7
Ce n’est que lorsqu’on tente de résoudre is(X,+(*(3,7),8)) que l'unification de X avec
I’évaluation du deuxieme opérande est effectuée. Cette évaluation est également commandée
par les opérations de comparaison : >=,=<,<,> ==\== les deux derniers dénotant 1’égalité ou
I'inégalité mais forcant I’évaluation des arguments.

1.5 Les listes

Dans nos exemples précédents, notre syntagme était représenté par un foncteur a n ar-
guments pour n mots, mais la longueur des suites de mots est inconnue a priori; il est donc
important de trouver une représentation pour exprimer des listes d’éléments. Ceci est fait
a I'aide d'un foncteur spécial . a deux arguments dont le premier indique un élément et le
deuxieme la suite de la liste. Donc la liste de deux éléments a et b est représentée comme
suit :

.(a,b)

cette notation est un peu encombrante et on utilise normalement ’abréviation infixe suivante
ou la liste est indiquée entre crochets et le début est séparé de la suite par une barre verticale.

[a | Db]
Ceci est particulierement pratique lorsqu’il y a plusieurs éléments dans la liste. Comparez :

(a, .(.(c,d),e)) et [a |[[cld]Ib]]
C.C.(c,d), e,a) et [[lcld]llb]lal

Pour éviter des niveaux de crochets, on peut remplacer . [ par une virgule (et enlever le
crochet correspondant). Par exemple le premier exemple devient :

[a, [cld] Ib]

La fin de liste est habituellement indiquée par [] et on permet de ne pas indiquer le dernier
élément vide (i.e.| [1 ); ceci permet de simplifier encore plus, comparez :

(a, .(b, .(c,[1))) [allbllcll1]]] [a,b,c]

qui représentent tous le méme objet; évidemment la troisieme possibilité est la plus pra-
tique a utiliser mais elle n’est qu'une abréviation pour les deux autres; la premiere étant
la représentation “logique”. La notation avec une barre verticale est habituellement utilisée
lorsqu’on veut indiquer que le “reste” d’une liste est une variable. Ainsi [a,b | X] équivaut
a .(a,.(®,X)) et représente une liste dont les deux premiers éléments sont a et b mais
dont on ne connait pas la suite (qui peut étre [1). [X|Y] équivaut & . (X,Y) et donc X est
le premier élément et Y le reste; ceci correspond au car et cdr de Lisp. Voyons mainte-
nant quelques prédicats utiles sur les listes. Nous n’en donnons que la définition et quelques
exemples d’utilisation. On pourra consulter (Clocksin et Mellish 84) (Sterling et Shapiro
86) pour apprendre a dériver ces prédicats. Nous allons les utiliser ici régulierement et les
supposer prédéfinis.

Toloth

oot outils de manipulations de listes
oloth fichier "listes.pro"

Tl to

b concatenation

concat ([],L,L).
concat ([X|Xs],Y, [X|Z]) :-concat(Xs,Y,Z).

pA verification d’appartenance
dans (X, [X|Xs]).
dans (X, [_|Xs]) :-dans(X,Xs) .

yA donner le nieme element
nieme([H|T],1,H):-!.
nieme([H|T] ,N,H1):-N1 is N-1, nieme(T,N1,H1).

A trouver le nombre d’elements
longueur([],0).

longueur ([H|T],N) :-longueur (T,N1) ,N is N1+1.

A renverser l’ordre des elements
renverse(L,R) :—-renverse(L, [],R).
renverse([H|T] ,L1,L) :-renverse(T, [H|L1],L).
renverse([],L,L).

Voyons maintenant quelques exemples d’utilisation de ces preédicats :

| ?- concat([1,e],[c,h,a,t],X).
X = [1,e,c,h,a,t]

| ?- concat(Radical, [e,r],[a,i,m,e,r]).
Radical = [a,i,m]

| ?- concat(Debut,Fin,[j,e,u,n,e]).
Debut = []
Fin = [j,e,u,n,e] ;

Debut = [j]
Fin = [e,u,n,e] ;

Debut = [j,e]
Fin = [u,n,e] ;

Debut = [j,e,ul
Fin = [n,e] ;

Debut = [j,e,u,n]
Fin = [e] ;

Debut = [j,e,u,n,e]
Fin = [] ;

no

| 7- dans(h, [c,h,a,t]).
yes
?- dans (X, [c,h,a,t]).

-

<> Ko<} —
|
B0

| 7- nieme([c,h,a,t],3,X).

X = a

| ?- longueur([c,h,a,t],L).
L=4

I
X

On remarque qu’on peut utiliser le méme prédicat de plusieurs facons différentes selon que
I’on instancie ou non certains parametres : ainsi concat sert aussi bien a coller deux listes
bout a bout qu’a en décomposer une en deux morceaux. dans permet d’obtenir tous les
éléments d’'une liste.

-~

- renverse([e,t,a,l,e,r],X).
[r,e,1l,a,t,e]

1.6 Prédicats extra-logiques

Si Prolog est basé sur la logique, on a quelquefois besoin d’un controle ou d’opérations qui
sortent du cadre de la logique. Nous allons ici décrire tres brievement les outils extra-logiques
dont nous aurons besoin pour le traitement de la langue naturelle.

1.6.1 Le coupe-choix

Si le non-déterminisme est souvent pratique, il y a des cas ol il est intéressant de le limiter.
L’outil le plus utile en Prolog est le “coupe-choix” cut notée par un point d’exclamation qui,
lorsqu’il est “prouvé”, a pour effet d’enlever tous les choix en attente dans le groupe de
clauses ou il apparait. Il est utilisé pour éviter des tests dont on est certain qu’ils vont
échouer ou pour s’assurer que certaines opérations ne vont étre effectuées qu’une seule fois.
Par exemple,

max(X,Y,X):—- X>=Y.

max(X,Y,Y):—- X<Y.
pourrait étre réécrit comme

max(X,Y,X):— X>=Y,!.

max(X,Y,Y).
Ainsi, si le test de la premiere clause réussit, il est inutile de tenter la deuxieme et s’il échoue,
alors il est stur que le deuxieme réussit et nous n’avons pas besoin d'un autre test.

Le coupe-choix permet également de forcer I’échec d’un prédicat méme s’il reste d’autres
possibilités a explorer. Un cas particulier mais tres utile de ce cas de figure est la définition
de not (P) qui réussit lorsque P échoue et échoue lorsque P réussit. Il est défini comme suit :

not(P) :- call(P), !, fail.

not (P) .
ou call est tout simplement une demande de résolution du prédicat P. Le coupe-choix assure
que la seconde clause ne sera pas tentée (pour réussir) apres le succes de P. Si P échoue, alors
le coupe-choix n’est pas essayé et on tente la deuxieme possibilité qui renvoie un succes car il
n’y a rien a prouver. En Sicstus, ce prédicat est prédéfini sous la forme de 'opérateur préfixe

\+.

1.6.2 Décomposition de littéraux

On a quelquefois besoin de manipuler les littéraux eux-mémes, en particulier lorsqu’on
calcule un but a résoudre. Comme en C-Prolog il n’est pas possible d’écrire le but X(Y) ou X
aurait été instancié dans la clause ou il apparait, il faut d’abord composer le foncteur avant
de I'appeler. Ceci peut étre fait en utilisant 'opérateur =. . (nommé univ depuis la premiere
version de Prolog). Ainsi

f(a,b,c) =.. X
unifiera X a [f,a,b,c]. On obtient donc une liste dont le premier élément est le foncteur et
les autres ses arguments. Donc pour construire X(Y,Z) et ’évaluer, on fera plutot

P=.. [X,Y,Z], call(P).

1.6.3 Décomposition d’un identificateur

On peut “exploser” ou “imploser” le nom d’'un identificateur avec name ; par exemple :
name (chat,X)

unifiera X a la liste [99,104,97,116] ou chaque élément correspond au code ASCII du
caractere correspondant de I'identificateur donné comme premier argument. Cette opération
est réversible en ce sens que si on donne une liste de codes ASCII comme deuxieme argument
et une variable comme premier, alors on recompose un identificateur. Comme la manipulation
de codes ASCII est assez malcommode et rend les programmes difficiles a lire, nous définirons
a la section suivante un autre prédicat nom qui ne travaille qu’en terme d’identificateur (mais
il utilise name de fagon interne).

1.6.4 Modification de la base de faits

On peut ajouter et retrancher dynamiquement des clauses a un programme : ainsi
assert(c(pl,p2)).

ajoute la clause c(pl,p2) au programme et
retract(c(X1,X2))

enleve la premiere clause qui s'unifie avec ¢ (X1,X2). Ce mécanisme est utilisé pour transfor-
mer des programmes ou pour conserver des informations entre plusieurs exécutions de clauses
car normalement toutes les liaisons de variables sont brisées lors de la fin de la preuve.

Nous donnons maintenant une illustration de I'arithmétique et de la modification de la
base de faits en définissant un prédicat permettant de tirer des nombres au hasard ; ceci nous
sera utile dans certains exemples des prochains chapitres.

oo ce qu’il faut pour generer au hasard
oo c’est le fichier "hasard.pro"

%  tirage au sort via un generateur congruentiel lineaire
A donne X un nombre reel dans [0,1]
hasard(X) :-

retract (germe_hasard(X1)), X2 is (X1%824) mod 10657,
assert(germe_hasard(X2)), X is X2/10657.0 .
% donne X un nombre entier dans [1,N]
hasard(N,X) :-
hasard(Y),
X is floor(NxY)+1.

% initialisation du germe (a la consultation du fichier)
:— abolish(germe_hasard,1), X is floor(cputime*1000),
assert (germe_hasard(X)).

1.7 Entrées-Sorties

Comme nous voulons traiter la langue naturelle et que ces applications impliquent souvent
I’écriture d’interfaces, il est important d’avoir des outils commodes d’entrée et de sortie
de mots et de phrases. Nous allons passer rapidement en revue les prédicats de lecture et
d’écriture de C-Prolog et ensuite nous montrerons la programmation de prédicats de plus
haut niveau.

1.7.1 Entrée de caracteres
Voici les prédicats permettant I'entrée de caracteres :

get0(X) unifie X avec le code ASCII du prochain caractere sur le terminal.
get (X) comme getO mais ignore tous les caracteres de controle.

skip(X) ou X doit étre instancié a un code ASCII, saute tous les caracteres en entrée en
s’arrétant au premier dont la valeur est égale a X.

1.7.2 Ecriture

Voici les prédicats permettant 1’écriture de caracteres :

put (X) écrit sur le terminal le caractere dont le code ASCII est X.
write(X) écrit le terme X selon la syntaxe standard y compris les opérateurs infixe
nl écrit une fin de ligne

tab(I) écrit I blancs sur la ligne courante. I peut étre une expression a évaluer.

1.7.3 Entrée de mots et de listes de mots

Nous définissons maintenant des prédicats qui permettent de lire une suite de mots séparés
par des blancs et terminée par un point, un point d’interrogation ou une fin de ligne et qui
retourne une liste d’identificateurs. Ainsi

| 7- lire_les_mots(X).

| : bonjour les degats

X = [bonjour,les,degats]

| ?7- lire_les_mots(X).

| : Salut, les grands copains
X = [Salut,les,grands,copains]

Ce programme n’est qu’'un exemple de traitement de listes de mots et il est facile de le modi-
fier pour traiter autrement certains caracteres. On y remarque également la transformation
des codes ASCII en identificateurs.

AN

%%%  Les utilitaires de lecture
oot fichier "entree.pro"

AN

%% transformation de la ligne lue en une suite d’atomes
lire_les_mots(Ws):-lire_car(C),lire_les_mots(C,Ws).

lire_les_mots(C, [W|Ws]) :- % ajoute un mot
car_mot(C),!,
lire_mot(C,W,Cl1),lire_les_mots(C1l,Ws).

lire_les_mots(C,[]):- % fin de phrase
car_fin_des_mots(C),!.
lire_les_mots(C,Ws):- % on oublie ce

lire_car(C1),lire_les_mots(C1l,Ws). % caractere

lire_mot(C,W,C1):- % construit un mot
cars_des_mots(C,Cs,C1) ,nom(W,Cs) .

cars_des_mots(C, [C|Cs],CO):-
car_mot(C),!,lire_car(Cl),cars_des_mots(C1,Cs,CO).
cars_des_mots(C, [1,C):— \+(car_mot(C)).

YAA classes des caracteres
car_mot(C):- a @=< C, C @=< z.
car_mot(C):-’A’ @=< C,C @=< ’Z’.

car_mot(’’’’).

oo une liste de mots est terminee par un point un

Yoo un point d’interrogation ou une fin de ligne
car_fin_des_mots(’.’) :-skip(10). % sauter jusqu’a la
car_fin_des_mots(’?7’) :-skip(10). % fin de ligne

car_fin_des_mots(X) :-name(X,[10]). % le "newline"

%kt  predicats necessaires en C-Prolog pour eviter
%%%  la manipulation de codes ascii

Yo lecture d’un caractere et
Yo transformation en 1’atome correspondant
lire_car(X) :- getO0(X1), name(X, [X1]).

ot nom explose un identificateur en

Dot une liste d’identificateurs d’une lettre

YA appel: nom(aimer, [a,i,m,e,r])

Dot

nom(X,X1) :- % explose
var(X1),!, name(X,Codes), noml(Codes,X1).

nom(X,X1) :- % compose

var(X),!, noml(Codes,X1l), name(X,Codes).

nom1 (], [1).
noml ([X|Xs], [X1]X1s]) :- name(X1, [X]), noml(Xs,Xls).

1.7.4 Ecriture de listes de mots et d’arbres

Prolog nous fournit les outils de base pour ’écriture de termes, mais dans une interface
nous aimerions souvent pouvoir écrire une liste d’identificateurs comme une liste de mots
séparés par un blanc et terminée par une fin de ligne. Le prédicat ecrire-les-mots permet
de faire cette opération. De plus, nous montrons un prédicat un peu plus complexe qui permet
de faire ressortir la structure d’un terme composé en utilisant I'indentation des différentes
lignes. Cette opération est communément appelé “pretty-print” d’ou le nom de pp pour ce
prédicat. Sa principale particularité étant qu’il est possible de controler les termes qui seront
écrits sur plusieurs lignes. Voici des exemples d’utilisation :

| ?- ecrire_les_mots([bonjour,les,degats]).
bonjour les degats

| 7- pp(ph(snm(det(le),nom(chat)),svb(mange,snm(det(la),nom(souris))))).
ph(snm(det (1e),

nom(chat)),

svb(mange,
snm(det (la),
nom(souris))))

Ces prédicats peuvent étre définis comme suit :
Dotolo
%hY%  Les utilitaires d’ecriture
oot fichier "sortie.pro"

Yototo

%% ecriture d’une liste de mots separes par un blanc

h% et terminee par une fin de ligne
ecrire_les_mots([H|T]) :—write(H) ,write(’ ’),ecrire_les_mots(T).
ecrire_les_mots([]) :- nl.

Tototo

ooto impression d’expressions Prolog avec controle

Yotote des foncteurs qui sont eclates sur plusieurs lignes

oot appel: pp(expression)

Yoot ou pp(expression,decalage)

oo to

pp(X) :-pp(X,0). % appel simplifie

pp(X,I):-var(X),!,write(’_’). % une variable 7

pp(X,I):-
X=.. [Y,Y1|Ys], boum(Y,Long),!, % predicat a eclater?
write(Y) ,write(’ (°), % ecrire le foncteur
pp(Y1,I+Long), % son premier argument
ppl(Ys,I+Long) ,write(’)’). % et les autres

pp(X,I) :-write(X) . % autre cas

oo to

oot impression d’une liste de parametres

oot sur des lignes differentes

Yoo en tenant compte du decalage

ooto

ppl([]1,I):-!.

ppl([X],I):-!,write(’,’),nl,tab(I),ppX,I).
ppl([X[Xs],I):-!,write(’,’),nl,tab(I),pp(X,I),ppl(Xs,I).

Yoo

Yoo les predicats a eclater avec le decalage

ooto des lignes suivantes normalement c’est 1 de plus
oot que la longueur de 1’identificateur

Yoo

boum(ph,3) .

boum(snm,4) .

boum(svb,4) .

boum(rel,4).

boum(int,4) .

Nous terminons ici notre courte présentation de Prolog. Elle est loin d’étre complete et
on pourra consulter les ouvrages cités en bibliographie. Les outils définis dans ce chapitre
serviront dans les prochains chapitres.



Chapitre 2

Analyse automatique en utilisant les
grammaires logiques

2.1 Introduction

Les grammaires présentées au chapitre précédent avaient pour but de décrire la structure
des phrases et certains phénomenes linguistiques. Il n’était pas nécessairement question de
développer un moyen automatique de mettre en adéquation une phrase et une structure.
Si nous voulons disposer d’outils informatiques de traitement de la langue naturelle, alors
nous avons besoin d’algorithmes permettant de déterminer automatiquement si une phrase
répond a une structure déterminée et, si oui, comment sont reliés les différents constituants
entre eux.

Les premiers travaux dans ce domaine ont été effectués dans le cadre des grammaires
formelles d’abord utilisées pour la définition de langages de programmation. Comme ce sont
des langages artificiels, il est possible de les fagconner sur mesure de fagon a obtenir des
analyseurs rapides et efficaces. Dans le cas de langues naturelles, il faut tenir compte de
phénomenes complexes et donc les structures des grammaires ou des analyseurs deviennent
plus compliquées.

Un formalisme qui a maintenant fait ses preuves est celui des grammaires logiques
basées essentiellement sur le langage Prolog. Son grand mérite est sa déclarativité : comme
nous l'avons vu au chapitre précédent, un programme Prolog est essentiellement une suite
d’énoncés logiques indiquant des relations entre ceux-ci et I'exécution d'un programme se
ramene a une preuve de la véracité d’un nouvel énoncé a partir de ceux qu’on a déja.

Par exemple, soit la structure suivante : une phrase (p) est composée d'un syntagme
nominal (sn) et d’un syntagme verbal (sv); un syntagme nominal (sn) est composé d'un
déterminant (det) et d'un nom (n); un syntagme verbal (sv) est composé d'un verbe (v)
ou d'un verbe et d'un syntagme nominal (sn). Ceci peut tres bien étre exprimé comme une
suite d’énoncés logiques décrits a la figure 2.1.

Maintenant, il s’agit de prouver si la phrase :

le chat mange la souris

correspond bien a notre définition d’une phrase correcte. Pour continuer notre analogie avec
la preuve de théoreme, il s’agit de montrer que la premiere relation est vraie pour cette

15

p = SnASsv
sn = detA\n
sv = vV (vAsn)

F1G. 2.1: Grammaire en forme logique

phrase : elle le sera si nous trouvons un sn et un sv ; un sv est composé d’un déterminant det
et d'un nom nom (ce qui correspond bien a le et a chat); il reste a prouver que la suite de
la phrase mange la souris est un sv composé d'un verbe v mange et d'un sn; la deuxieme
relation nous indique encore que le déterminant la et le nom souris forment un tel syntagme.
Nous avons donc ainsi réussi a “prouver” que notre phrase a une structure répondant bien
a notre définition. Evidemment, il faut remarquer que le et (A) est assez spécial car en plus
d’indiquer la conjonction, il indique la juxtaposition ; il n’est donc pas commutatif. On peut
représenter ce traitement avec le schéma suivant :

| le chat | mange la souris |
| <= sn -> | <—————- SV ————---- > |

| <= v => | <== sn -—> |
Ce petit exemple informel illustre I'intuition derriere 1'utilisation de la programmation lo-
gique (ou d’un démonstrateur de théoréme) pour I'analyse de langages. Evidemment, il reste
encore plusieurs problemes a régler et nous allons regarder les outils fournis par Prolog pour
exprimer une gramimaire.

2.2 Codage de la grammaire en Prolog

Maintenant voyons comment coder la grammaire de la figure2.1 en Prolog, de fagon a
vérifier si une phrase donnée peut étre analysée par cette grammaire. La premiere décision
concerne évidemment la représentation de la phrase a analyser. Nous prenons la représentation
la plus simple et la plus intuitive : une phrase est une suite de mots considérés comme ato-
miques. Ainsi la phrase
Le chat mange la souris.
sera représentée comme une liste d’atomes Prolog.

[le,chat,mange,la,souris]

Normalement c’est une interface qui construit cette liste par I’emploi d’outils semblables au
prédicat lire_les_mots que nous avons défini au chapitre 2. Il reste maintenant a coder les
regles : nous choisissons la représentation suivante : un symbole non-terminal sera représenté
par un prédicat a deux arguments : le premier indique la chaine d’entrée et le second,
ce qu’il reste de la chaine apres son traitement par le prédicat. Cette représentation est
communément appelée liste de différences. (Pereira Shieber 87) (Giannesini 85) montrent
comment il possible de dériver “naturellement” cette représentation. Ici nous la prenons

tout simplement comme acquise et elle date d’ailleurs des premiers travaux de (Colmerauer
78) sur les grammaires de métamorphose. La définition d'un syntagme verbal peut donc étre
exprimée ainsi :

sv(LO,L) :- v(LO,L).
sv(LO,L) :- v(LO,L1), sn(L1,L).

Les parametres Prolog nous permettent de connecter les différentes listes : dans le premier
cas, ou il n’y a qu’'un verbe pour le syntagme, les listes du syntagme sont évidemment celles
du verbe. Dans le deuxieme cas, la liste d’entrée du verbe est celle du syntagme mais sa
sortie est donnée en entrée au syntagme nominal pour qu’il fasse ’analyse de la suite et sa
propre sortie sera celle du syntagme.

Dans le cas d’'un symbole terminal, il suffit d’avoir une clause qui vérifie la présence du
terminal au début de la chaine et qui donne en sortie la suite de la chaine. Cette vérification
s’exprime simplement avec la clause unitaire suivante :

terminal (Mot, [Mot | L], L).

Son utilisation est également naturelle : par exemple, pour vérifier qu'un nom peut étre chat,
on écrit la clause :

n(LO,L) :- terminal(chat,LO,L).

un lecteur attentif ou un programmeur Prolog averti aura tout de suite remarqué qu’il est
possible d’optimiser cette clause en intégrant immédiatement la clause terminal. On obtient
alors par substitution de Mot et L :

n([chat | L], L)

mais, par souci d'uniformité et aussi parce que nous verrons que ce sera le systeme qui
va gérer ces listes d’entrée et de sortie, nous garderons la premiere version.

La grammaire complete devient donc :
otolo
Tototo Grammaire en Prolog
Dooth (grammairel.pro)
oot
p(LO,L) :- sn(LO,L1), sv(L1,L).

sn(LO,L) :- det(LO,L1), n(L1,L).

sv(LO,L) :- v(LO,L).
sv(LO,L) :- v(LO,L1), sn(L1,L).

det(LO,L) :- terminal(le,LO,L).
det(LO,L) :- terminal(la,LO,L).

n(LO,L) :- terminal(souris,LO,L).
n(LO,L) :- terminal(chat,LO,L).

v(LO,L) :- terminal (mange,LO,L).

v(LO,L) :- terminal(trottine,LO,L).

terminal (Mot, [Mot|L],L).
que nous pouvons tester comme suit :

| ?- p([le,chat,mange,la,souris],[]).
yes
| 7- p([le,chat,mange], []).
yes
| 7- p([la,chat,mange,le,souris],[]).
yes
| 7- p([le,chat,trottine,la,souris],[]).
yes
| 7- p([lelX],[1),write([1lelX]),nl,fail.
[le,souris,mange]
[le,souris,trottine]
[le,souris,mange,le,souris]
[le,souris,mange,le,chat]
[le,souris,mange,la,souris]
[le,souris,mange,la,chat]
[le,souris,trottine,le,souris]
[le,souris,trottine,le,chat]
[le,souris,trottine,la,souris]
[le,souris,trottine,la,chat]
[le,chat,mange]
[le,chat,trottine]
[le,chat,mange,le,souris]
[le,chat,mange,le,chat]
[le,chat,mange,la,souris]
[le,chat,mange,la,chat]
[le,chat,trottine,le,souris]
[le,chat,trottine,le,chat]
[le,chat,trottine,la,souris]
[le,chat,trottine,la,chat]
no

Comme le montrent les troisieme et quatrieme exemples, notre grammaire naive accepte des
phrases ayant une structure correcte, mais la langue naturelle impose d’autres contraintes
qu’il faut vérifier : par exemple, ’accord en genre entre ’article et le nom et le fait que seul un
verbe transitif peut accepter un complément d’objet direct. Ce “laxisme” est particulierement
évident dans le dernier exemple qui nous engendre toutes les phrases possibles commencant
par le a partir de la grammaire; il faut tout de suite remarquer que si cette exploitation “a
I'envers” de la grammaire est un avantage indéniable de 'utilisation du non-déterminisme de
Prolog, cette méthode n’est pas vraiment utilisable en pratique comme outil de génération
de texte; en effet, une grammaire réelle générera trop de possibilités et aura souvent une
structure qui bouclerait a la génération, sans compter que la génération de texte est beaucoup

plus qu’une simple sortie de listes de mots (voir chapitre 6).

Voyons maintenant comment implanter ces vérifications : il faut tout d’abord coder I'in-
formation sur le genre des articles et des noms et sur la transitivité des verbes. L’approche
la plus simple est de la coder dans la grammaire et de la répercuter via des parametres qui
devront s'unifier pour assurer la conformité des informations, notre grammaire devient donc :
oot
oot Grammaire avec vérifications
oloth (grammaire2.pro)
oot
p(LO,L) :- sn(LO,L1), sv(L1,L).

sn(LO,L) :- det(Genre,LO,L1), n(Genre,L1,L).

sv(LO,L) :- v(_,LO,L).
sv(LO,L) :- v(tramnsitif,LO,L1), sn(L1,L).

det (masculin,LO,L) :- terminal(le,LO,L).
det(feminin,LO,L) :- terminal(la,LO,L).

n(feminin,LO,L) :- terminal(souris,LO,L).
n(masculin,LO,L) :- terminal(chat,LO,L).
v(transitif,LO,L) :- terminal(mange,LO,L).
v(intransitif,LO,L) :- terminal (trottine,LO,L).

terminal (Mot, [Mot|L],L).

L’information de base est donnée comme premier parametre aux prédicats det, n et v.
La vérification de conformité de genre est faite dans la clause sn qui force le déterminant
et le nom a avoir la méme valeur pour Genre. La transitivité est vérifiée dans sv : dans la
premiere clause, nous décidons ici d’accepter un verbe transitif ou intransitif comme verbe
sans complément et nous donnons tout simplement une variable muette notée par un souligné
(_); dans la deuxieme clause de sv, nous exigeons toutefois la caractéristique transitif.

Testons maintenant cette grammaire sur les cas problemes pour constater qu’ils sont
maintenant rejetés :

| ?- p([la,chat,mange,le,souris],[]).
no

| ?- p([le,chat,trottine,la,souris], []).
no

Jusqu’ici, nous nous sommes limités a vérifier si une phrase répondait a une structure,
mais nous aimerions que notre processus d’analyse nous permet-te d’obtenir la structure de
dépendance entre les constituants de notre phrase d’entrée. Voyons maintenant comment
Prolog nous permet de construire ces structures, ici encore la procédure d’unification de
Prolog nous donne un outil puissant, simple et naturel de construire de fagon incrémentielle

la structure d’une phrase. Nous rappelons qu'une structure arborescente peut étre représentée
facilement en Prolog avec un foncteur indiquant la valeur du noeud et dont les parametres
corespondent aux branches issues de ce noeud. Ainsi I’arbre :

Tt X
le chat mange ﬂt nfﬂ

la  souris
sera représenté par

ph(snm(dt(le),
nm(chat)),
svb (vb (mange) ,
snm(dt (la),
nm(souris))))

Cet arbre est construit via un parametre véhiculant I'information structurelle. Un foncteur
Prolog est tres facile a construire car il suffit d’écrire sa forme et le processus d’'unification le
complete ou vérifie son adéquation. Ainsi la structure d’une phrase est donnée par le foncteur
ph avec comme parametre la structure du syntagme nominal et celle du syntagme verbal :

p(ph(SN_Struct,SV_Struct) ,LO,L) :-
sn(SN_Struct,LO0,L1), sv(SV_Struct,L1,L),

Nous faisons de méme pour le sn et le sv ainsi que pour les terminaux. Notre grammaire
n’est maintenant constituée que des prédicats d’analyse mais rien ne nous empéche d’uti-
liser d’autres prédicats Prolog pour vérifier des conditions supplémentaires (par exemple
sémantiques) qui demandent d’autres opérations que la vérification de conformité disponible
par 'unification. Par exemple, pour éviter ’acceptation de la phrase :

La souris mange la souris

il faudra vérifier qu’on ne retrouve pas le méme syntagme nominal avant et apres le verbe.
Nous pouvons ajouter ce test dans p pour s’assurer qu’on ne retrouve pas la méme structure
de syntagme nominal comme sujet et comme complément du verbe. Pour refuser une phrase
comme

La souris mange le chat.

Il faut vérifier que celui qui est mangé par un autre est bien une proie potentielle... On
ajoutera donc de l'information “sémantique” a notre programme et on la vérifiera dans la
clause sv. Notre nouvelle grammaire devient donc :

T th
oot grammaire avec construction de structures
oot (grammaire3.pro)

Yototo

p(ph(SN_Struct,SV_Struct),LO,L) :-
sn(SN_Struct,LO,L1), sv(SV_Struct,L1,L),
% s’assurer que les SN avant et apres le verbe sont differents
\+(SV_Struct = svb(_,SN_Struct)).

sn(snm(Det_Struct,N_Struct),LO,L) :-
det (Det_Struct,Genre,L0,L1), n(N_Struct,Genre,L1,L).

sv(svb(vb(Mot)),LO,L) :— v(vb(Mot,_),_,LO,L).
sv(svb(vb(Mot) ,SN_Struct),LO,L) :-
v(vb(Mot,Comp) ,transitif,LO,L1), sn(SN_Struct,L1,L),
% verifier que le complement est semantiquement acceptable

SN_Struct = snm(_,nm(Nom)), % extrait le nom
T =.. [Comp,Nom], % construit le predicat

% de verification semantique
call(T). % on 1’appelle

det(dt(le) ,masculin,lO,L) :- terminal(le,LO,L).
det(dt(la),feminin,L0,L) :- terminal(la,LO,L).

n(nm(souris) ,feminin,LO0,L) :- terminal (souris,LO,L).
n(nm(chat) ,masculin,LO,L) :- terminal(chat,LO,L).
v(vb(mange,proie) ,transitif,LO,L) :- terminal (mange,LO,L).
v(vb(trottine,_),intransitif,L0,L) :- terminal(trottine,LO,L).

terminal (Mot, [Mot|L],L).

% la semantique
proie(souris).

Evidemment nous pourrions continuer ainsi a ajouter des informations de toutes sortes et,
si nous regardons bien, nous constatons que 'expression de la grammaire est assez naturelle :
nous ajoutons des parametres pour véhiculer 'information entre les noeuds de 'arbre. 1l reste
toutefois a gérer le passage des listes d’entrée et de sortie entre les prédicats d’analyse d'une
méme clause. Ici notre grammaire est tres simple et il est facile de le faire a la main; or
cet ajout est systématique et pourrait étre facilement automatisé : c¢’est d’ailleurs ce qui a
été fait pour les grammaires de métamorphose (Colmerauer 78) et plus récemment pour les
grammaires a clauses définies (Definite Clause Grammars ou DCG) (Pereira et Warren 80)
qui en sont un cas particulier. Voyons maintenant notre derniere grammaire écrite avec le
formalisme DCG ou les parties gauches sont séparées des parties droites par (-->). Dans
ces clauses, c’est le systeme qui gere le passage de la chaine entre les prédicats; en fait,
dans beaucoup de systemes Prolog, ces clauses sont transformées a la lecture en des clauses
équivalentes ayant une structure semblable a celle que nous avons développée au cours des
exemples précédents, c’est-a-dire avec deux parametres supplémentaires a tous les prédicats

d’analyse ; mais ils ne doivent pas étre ajoutés aux autres prédicats que nous devons marquer
spécialement. En DCG, ces prédicats sont simplement entourés d’accolades. Un terminal est
indiqué entre crochets et la clause terminal n’est plus nécessaire dans notre programme car
elle est maintenant prédéfinie par le systeme (elle est traditionnellement appelée >C’ pour
Connects).

oo

oot Version DCG de la grammaire avec construction
ooto de structures

oo lo (grammaire3dcg.pro)

Yoo

p(ph(SN_Struct,SV_Struct)) -->
sn(SN_Struct), sv(SV_Struct),
% s’assurer que les SN avant et aprés le verbe sont différents
{not (SV_Struct = svb(_,SN_Struct))}.

sn(snm(Det_Struct,N_Struct)) -—>
det (Det_Struct,Genre), n(N_Struct,Genre).

sv(svb(vb(Mot))) --> v(vb(Mot,_),_).

sv(svb(vb(Mot) ,SN_Struct)) -—>
v(vb(Mot,Comp) ,transitif), sn(SN_Struct),
% verifier que le complément est sémantiquement acceptable
{SN_Struct = snm(_,nm(Nom)), (T =.. [Comp,Nom]), call(T)}.

det(dt(le) ,masculin) ——> [le].
det(dt(la) ,feminin) --> [la].

n(nm(souris) ,feminin) --> [souris].
n(am(chat) ,masculin) --> [chat].

v(vb(mange,proie) ,transitif) --> [mange].
v(vb(trottine,_),intransitif) --> [trottine].

% la semantique
proie(souris).

Une grammaire DCG est donc présentée sous la forme de regles de grammaire exprimée
par des prédicats avec des parametres permettant de véhiculer de 'information entre eux.
De plus, il est possible d’ajouter des prédicats Prolog quelconques a condition de les entourer
d’accolades.

Le formalisme DCG est a priori indépendant de Prolog et nous pourrions en écrire des
interpretes ou des compilateurs dans tout langage ou systeme qui permettrait d’unifier des
parametres. Evidemment comme ce sont des opérations fondamentales de tout interprete
Prolog, les DCG sont assez simples a implanter dans ce langage car il suffit d’un simple
préprocesseur (lui-méme écrit en Prolog).

En comparant les deux derniers programmes, on constate que le fait de cacher les mani-
pulations de listes permet de faire ressortir les traitements et les parametres importants; les
accolades regroupent les traitements autres que I’analyse syntaxique. C’est pourquoi a partir
de maintenant nous n’utiliserons que le formalisme DCG; mais si votre Prolog favori ne
dispose pas d'un formalisme équivalent, il vous suffira d’ajouter a la main les listes d’entrées
et de sorties' pour appliquer ce que nous montrons ici.

2.3 Une autre application des DCG

Nous allons nous servir des DCG pour écrire une version de Eliza, un programme écrit
originalement par J. Weizenbaum au milieu des années soixante pour illustrer un systeme
de traitement de liste. Comme ce programme avait pour cadre la conversation entre un
psychanalyste et son patient, il a excité 'intérét du public non averti. En y regardant de
plus pres, on constate que ce programme n’est constitué que d’une suite de modeles cherchés
dans une chaine d’entrée qui sont répétés avec quelques modifications mineures. Ceci peut
tres bien étre modélisé en terme de grammaires et c’est pourquoi nous allons utiliser les
DCG pour les exprimer. Ici au lieu de construire ’arbre de dérivation lors de I'analyse d’une
chaine, nous allons tout simplement donner la “réponse” du psychanalyste...

Cette grammaire est d’ailleurs tres simpliste ici car elle se borne a une suite de termi-
naux a trouver dans l’entrée. La seule complication consiste a permettre la recherche de ces
terminaux a partir de n’importe quelle position dans la chaine d’entrée. Pour cela, il suffit de
pouvoir exprimer une chaine de longueur “variable” dans l’entrée. Ceci peut étre facilement
exprimé en utilisant le non-déterminisme implicite de Prolog avec le prédicat concat défini
au chapitre précédent. Pour bien montrer que ce qui nous intéresse est plutot le fait de sauter
des parties de chaine nous nommons ce prédicat “...” ; ¢’est un prédicat a trois arguments,
mais comme les deux derniers sont fournis par les DCG, c¢’est donc un prédicat a un seul
argument ; nous pouvons donc nous passer des parentheses autour de son parametre en le
déclarant comme opérateur préfixe. Ainsi . . . _ désigne une liste éventuellement vide de mots
qui sont ignorés et ...X unifie X avec la liste de mots qui sont sautés. Afin d’introduire un
peu de variation, nous utilisons le prédicat, ok (P) qui ne réussit qu’avec une probabilité de P
en utilisant le prédicat hasard défini dans un fichier que nous décrivons pas ici; hasard(X)
retourne un nombre au hasard différent dans I'intervalle [0, 1) a chaque appel.

oot programme Eliza de Weizenbaum qui simule une
oot conversation avec un thérapeute.

Tototo Le programme original a été fourni par Michel Boyer.
Totoo Les formules frangaises sont inspirées de

VAN Wertz, Lisp, Masson, 1985, p180

otole (eliza.pro)

:— reconsult(’entree.pro’), reconsult(’sortie.pro’),
reconsult(’hasard.pro’), reconsult(’listes.pro’).

lou consulter (Sterling et Shapiro 86), (Pereira et Sheiber 87) ou (Saint-Dizier 87) (en frangais), pour

écrire un interprete ou un traducteur qui le fera automatiquement

% autre nom pour "concat" pour indiquer que tout ce qui
% mnous interesse ici c’est de sauter un bout de phrase
%» attention le resultat est dans le deuxieme parametre !!!
:= op(900,fy,’...").
...(X,Y,Z) :- concat(X,Z,Y).

eliza :-
write(’ Oui, tu peux tout me dire;), nl,
write(’|: ?),
repeat,
lire_les_mots(Entree),
eliza(Entree).

eliza([bonsoir]) :-
write(merci), nl, !.
eliza(Entree) :-
reponse (Reponse,Entree, []1), !,
ecrire_les_mots(Reponse), fail.

%% Generation de la reponse a une entree du patient.
%% reponse(LaReponse) --> le modele a verifier dans 1l’entree
reponse ([pourquoi,n,etes,vous,pas | X]) -->

[je,ne,suis,pas], ...X.

reponse([alors,vous,savez,programmer]) -->
.._, [prologl, ..._.
reponse([alors,vous,ne,savez,pas,programmer]) -->
_, [lispl, ..._.

reponse([dites,m,en,plus,sur,votre,X]) -->

cee_, [X1, ..._, {important(X)}.
reponse ([hmmm]) -->

_,_,_1.
reponse([vous,etes,bien,negatif]) -->

[non] .

reponse([c,est,un,peu,court]) -->
[_], {ok(0.33)%.
reponse([vous,n,etes,pas,bavard]) -->
[_1, {ok(0.5)}.
reponse([vous,m,en,direz,tant]) -->
(1.
%» on n’a pas trouve les modeles qu’on cherchait
reponse([je,ne,vous,suis,pas,tres,bien]) -->
.._,{0k(0.33)%}.
reponse([ca,alors]) -->
..._,{0k(0.5)}.

reponse([n,importe,quoi]) -->

%» ne reussit qu’avec probabilite P
ok(P) :— hasard(X),!,X<P.

%% la semantique
important (mere) .
important (pere) .
et un exemple d’interaction avec ce programme
| 7- eliza.
OQui, tu peux tout me dire!
|: j’aime bien faire du prolog
alors vous savez programmer
|: oui
vous n etes pas bavard
| : non
vous etes bien negatif
| : pas autant que mon pere
dites m en plus sur votre pere
|: je ne suis pas tout a fait votre conversation!
pourquoi n etes vous pas tout a fait votre conversation
| : parce que ce serait trop long
n importe quoi
| : vous m insultez
hmmm
|: c’est tout ce que ca vous fait
n importe quoi
|: eh oui
ca alors
| : bonsoir.
merci

L’adaptation francaise de ce programme est moins spectaculaire que la version originale
anglaise car il faut trouver des formules invariantes sous le genre et le nombre tout en tenant
compte des variations morphologiques. Ce probleme étant beaucoup moins aigu en anglais, il
était plus facile de tromper l'interlocuteur avec un programme plus simple ne faisant somme
toute que des remplacements ou des déplacements de mots rencontrés précédemment dans
les questions. On pourrait arriver au méme résultat en francgais en utilisant une program-
mation plus complexe pour tenir compte de la morphologie mais ceci n’ajouterait rien a la
compréhension.

Nous vous invitons a comparer cette version avec ce qu’il aurait fallu écrire dans d’autres
langages notamment Lisp (utilisé habituellement pour programmer cet exemple); il faut
alors écrire un algorithme de vérification de modele assez complexe alors qu’ici on utilise
tout simplement 1'unificateur de Prolog.



